

```
1 ' PROGRAM (1)
2 ' load and save
3 ' headerless files
4 ' by B.T. 2.4.1987
5 ' (c) PIXEL
10 w$(0)="LOAD":w$(1)="SAVE"
20 INPUT"Enter 0/LOAD 1/SAVE";sav
30 PRINT"start address";:GOSUB 280:addr$=s$
40 IF VAL("&"+addr$)<&A67C THEN MEMORY addr-1
50 PRINT"length";:GOSUB 280:length$=s$
60 INPUT"sync. code";code
70 low=VAL("&"+RIGHT$(addr$,2))
80 high=VAL("&"+LEFT$(addr$,2))
90 low1=VAL("&"+RIGHT$(length$,2))
100 high1=VAL("&"+LEFT$(length$,2))
110 ps=&BE00:RESTORE 150
120 READ cod$:IF cod$="end" THEN 170
130 POKE ps,VAL("&"+cod$)
140 ps=ps+1:GOTO 120
150 DATA 21,00,00,11,00,00,3e,00
160 DATA cd,a1,bc,d8,32,15,be,c9,end
170 POKE &BE15,&FF:POKE &BE01,low
180 POKE &BE02,high:POKE &BE04,low1
190 POKE &BE05,high1:POKE &BE07,code
200 IF sav=1 THEN POKE &BE09,&9E
210 CALL &BE00:check=PEEK(&BE15)
220 IF check<>&FF THEN 240
230 PRINT w$(sav);" file - O.K.":END
240 IF check=0 THEN PRINT"ESCAPE pressed":STOP
250 IF check=1 THEN PRINT w$(sav);" error, try again":STOP
260 IF check=2 THEN PRINT"Check failure":STOP
270 PRINT"Something wrong":END
280 INPUT s:s$=HEX$(s,4):RETURN
```

```
1 ' PROGRAM (2)
2 ' make headers
3 ' for any file
4 ' by B.T. E.S.87
5 ' (c) PIXEL
```

```

10 INPUT"Give filename (16 char.)";name$
20 IF LEN(name$)>16 THEN 10
30 INPUT"Block number (0-255)";bn
40 IF bn>255 OR bn<0 THEN 30
50 INPUT"Last block (Y/N)";lb$:lb$=UPPER$(lb$)
60 IF lb$<>"Y" AND lb$<>"N" THEN 50
70 INPUT"File type (0-255)";ft
80 IF ft>255 OR ft<0 THEN 70
90 INPUT"Whole data length (0-65535)";d1
100 IF d1>65535 OR d1<0 THEN 90
110 INPUT"Start address (0-65535)";sa
120 IF sa<0 OR sa>65535 THEN 110
130 INPUT"First block (Y/N)";fb$:fb$=UPPER$(fb$)
140 IF fb$<>"Y" AND fb$<>"N" THEN 130
150 INPUT"Number of recorded bytes (0-65535)";rb
160 IF rb>65535 OR rb<0 THEN 150
170 INPUT"Execution address (0-65535)";ea
180 IF ea<0 OR ea>65535 THEN 170
190 INPUT"Enter any string (26 chrs.) as password";ps$
200 IF LEN(ps$)>26 THEN 190
210 INPUT"Continue (Y/N)";c$:c$=UPPER$(c$)
220 IF c$<>"Y" AND c$<>"N" THEN 210
230 IF c$="N" THEN MODE 1:RUN
240 MEMORY &A40F
250 addr=&A410:RESTORE 290
260 READ cod$:IF cod$="end" THEN 300
270 POKE addr,VAL("&" +cod$)
280 addr=addr+1:GOTO 260
290 DATA 3e,2c,21,00,a5,11,40,00,cd,9e,bc,c9,00,00,end
300 addr=&A500:FOR t=addr TO addr+64:POKE t,0:NEXT
310 FOR t=1 TO LEN(name$)
320 POKE addr+t-1,ASC(UPPER$(MID$(name$,t,1)))
330 NEXT
340 POKE addr+16,bn
350 POKE addr+17,ABS(lb$="Y")
360 POKE addr+18,ft
370 POKE addr+19,VAL("&" +RIGHT$(HEX$(d1,4),2))
380 POKE addr+20,INT(d1/256)
390 POKE addr+21,VAL("&" +RIGHT$(HEX$(sa,4),2))
400 POKE addr+22,INT(sa/256)
410 POKE addr+23,ABS(fb$="Y")
420 POKE addr+24,VAL("&" +RIGHT$(HEX$(rb,4),2))
430 POKE addr+25,INT(rb/256)
440 POKE addr+26,VAL("&" +RIGHT$(HEX$(ea,4),2))
450 POKE addr+27,INT(ea/256)
460 FOR t=1 TO LEN(ps$)
470 POKE addr+27+t,ASC(UPPER$(MID$(ps$,t,1)))
480 NEXT
490 PRINT"Press any key to save header"
500 CALL &BB18:CALL &A410:END

```

```
1 'PROGRAM (3)
2 'header-reader
3 'for tape only
4 'by B.T. 12.4.87
5 '(c) PIXEL
10 MODE 1:CALL &BBFF:CALL &BB4E:MEMORY 39999:RESTORE 30
20 FOR a=&A028 TO &A039:READ b$:POKE a,VAL("&" + b$):NEXT
30 DATA 3e,2c,21,40,9c,11,40,00,cd,a1,bc,d8,c6,fd,32,04,a6,c9
40 CLS:PRINT"    Press any key";CHR$(7):CALL &BB18:PRINT:PRINT
50 POKE 42500,0:CALL 41000
60 IF PEEK(42500)=253 THEN PRINT"ESCAPE pressed...":GOTO 370
70 IF PEEK(42500)<>0 THEN PRINT"Tape Error":PRINT:GOTO 370
80 CLS:PRINT"Tape Header Reader :":PRINT
90 PRINT:PRINT"FILENAME : ";:tw=0
100 FOR a=40000 TO 40015:PRINT CHR$(PEEK(a));
110 tw=tw+PEEK(a):NEXT
120 IF tw=0 THEN PRINT"Unnamed"
130 PRINT:PRINT:PRINT"Block Number :";PEEK(40016)
140 PRINT:PRINT"Last Block ? : ";
150 IF PEEK(40017)=0 THEN PRINT"No." ELSE PRINT"Yes."
160 PRINT:PRINT"File TYPE : ";:a=PEEK(40018)
170 a$=BIN$(a,8):typ=VAL(MID$(a$,5,3))
180 IF typ=0 THEN PRINT"Basic program"
190 IF typ=1 THEN PRINT"Binary File"
200 IF typ=2 THEN PRINT"Screen Image"
210 IF typ=3 THEN PRINT"ASCII"
```

```
220 IF typ>3 THEN PRINT"Unknown"
230 PRINT:PRINT"Protected ? : ";
240 IF VAL(RIGHT$(a$,1))=1 THEN PRINT"Yes" ELSE PRINT"No"
250 dat=PEEK(40019)+256*PEEK(40020)
260 PRINT:PRINT"Recorded bytes : ";dat
270 adr=PEEK(40021)+256*PEEK(40022)
280 PRINT:PRINT"Start adress :";adr
290 PRINT:PRINT"First Block ? : ";
300 IF PEEK(40023)=0 THEN PRINT"No" ELSE PRINT"YES"
310 dt=PEEK(40024)+256*PEEK(40025)
320 PRINT:PRINT"Whole data length :";dt
330 ea=PEEK(40026)+256*PEEK(40027)
340 PRINT:PRINT"Execution adress : ";
350 IF ea<>0 THEN PRINT ea ELSE PRINT"Not given"
360 LOCATE 1,25:PRINT "Continue...? (Y/N)"
370 k$=UPPER$(INKEY$):IF k$="" THEN 370
380 IF k$="Y" THEN 50
390 IF k$<>"N" THEN 370
400 END
```